
1.0 Executive Summary 
 
A change to the TF coil conductor and winding pack (WP) turn array has lead to an update of the 
smeared material properties for use in global models. This memo presents an analysis of this latest 
configuration, which includes the recent addition of a conductor-to-turn-wrap slip-plane (Kapton-
wrap). 
 
A parametric, text-based ANSYS1 input file is developed to represent the coil cross-section. Loads 
and boundary conditions are applied to simulate unit loads that lead to elastic moduli (Ex, Ey, Ez), 
Poisson’s Ratios (νxy, νyz, νxz), shear moduli (Gxy, Gyz, Gxz) and coefficients of thermal expansion 
(CTEs αx, αy, αz). The model is exercised with the latest dimensions (e.g. 0.966” x 0.709” 
conductor, a 49 mil Kapton-glass turn-wrap insulation, and a 0.120” thick glass ground wrap). In 
addition, the model is used to produces smeared properties for thicker ground wraps (GW), as this 
may be helpful to the no-case TF coil configuration. However, the attached input file can easily be 
rerun with different dimensions and the revised properties gleaned from the titles of automated 
result plots. 
 
A concern over the adequacy of these developed properties to capture the behavior of gross bending 
leads to the development of two “beam” models: 
 

• A detailed model of the TF WP with all of its conductors, slip-planes, insulation and GW. 
• A homogeneous model of the TF WP with orthotropic properties. 

 
The analysis of the detailed beam provides a benchmark of what bending deformations aught to be 
like. The analysis of the homogeneous beam with orthotropic properties is expected to produce 
similar results. However, the smeared model highlights a shortcoming in the smeared property 
calculation; the high transverse shear modulus is more appropriate for short beam shear and not the 
larger-scale bending deformations expected in our TF coil. Fortunately, applying the lower in-plane 
shear modulus in all three shear directions leads to deformations which are fairly consistent with the 
results predicted by the detailed model. All ANSYS input files are attached, along with text and 
figures to illustrate this point. 

                                                 
1 ANSYS Release 8.1, UP20040329, INTEL NT, ANSYS, Inc., Canonsburg, PA. 

2.0 Assumptions and Notable Concerns 
 
2.1 The analyses assume a zero-friction Cu-insulation interface. The real presence of friction (or 
partial and poor bonding) would effectively stiffen the moduli, pushing them closer to the earlier 
monolithic results. Therefore, the zero-friction assumption should yield conservative results (i.e., 
somewhat larger than expected deformations) when used in global model analyses. 
 
2.2 The orthotropic properties used for the Kapton-S2 glass layer are not necessarily based on the 
exact lay-up of our TF turn insulation and epoxy. However, they seem to be the next-best thing 
short of a costly and time-consuming test program. (See Attachment 6.2) 

3.0 Analysis 
 
The 3D structural ANSYS model is based on the TF coil conductor and winding pack shown in 
Attachment 6.1. The 0.966” x 0.709” Cu conductor has a 0.312” diameter cooling channel, and is 
wound into a three radial layer by four toroidal pancake array. Each turn is insulated with a single 
half-lap Kapton layer (3.5 mils with adhesive) and three half-laps of S-2 glass (7 mils each), for a 
total thickness of 2x3.5+6x7 or 49 mils. The entire WP is covered by a multiple layer glass wrap 
with a total thickness of 0.12” (in this model plot), although this is increased to 3/8” to add a case-
like characteristic to the structure. 
 
A representative 3D model is shown in Fig. 3.0-1. The local blow-up exposes an important model 
feature; the Cu turns are separated from the turn insulation by a small (10 μm) gap and series of 
simple gap elements (CONTAC52). The node-to-node contact formulation requires that the contact 
surfaces are normal to the line element, or defined by unit vectors (NX, NY, NZ). Since controlling 
the location of each node pair is not a simple task, the later approach is applied. The somewhat 
skewed view also shows the depth of the model. 
 
The constituent material properties are tabulated in Attachment 6.2. These properties are taken from 
various NBS, ITER and NCSX test data, as detailed in the referenced document. Unit loads are 
applied to the model to capture the essential behaviors leading to orthotropic moduli, CTEs and 
Poisson’s ratios. These results are discussed in the following section. 

Fig. 3.0-1 Annotated Model Plot (x points right, y points up and z is out of the page) 

4.0 Results 
 
The results of the analyses are summarized in Table 4.0-1 and illustrated by a series of 
representative plots. For efficiency, only the plots of the 0.25” thick ground wrap are presented. 
 
 

Table 4.0-1 Smeared Material Property Summary for Various GW Thicknesses 
Material Property 

Model and Global CS 0.12” GW 0.25” GW 0.375” GW 0.500” GW 

Ey (Et), [GPa] 51.0 45.1 41.2 38.3 
Ex (Er), [GPa] 53.2 46.6 42.3 39.1 
Ez (Ep), [GPa] 91.9 83.3 76.6 71.0 
νxy (νrt), [N/A] 0.390 0.342 0.310 0.285 

νyx (νtr), [N/A] 0.435 0.366 0.328 0.301 

AVE[νxz, νyz)] (νrp), [N/A] 0.304 0.294 0.284 0.275 

αx (αr), [μ/K] 8.1 8.8 9.5 9.9 

αy (αt) , [μ/K] 8.3 9.1 9.6 10.1 

αz (αp) , [μ/K] 13.4 13.2 13.0 12.7 
Gzx (Gpr), [GPa] 33.1 29.7 27.1 24.9 
Gzy (Gpt), [GPa] 33.0 29.7 27.1 24.9 
Gxy (Grt), [GPa] 2.1 2.1 2.2 2.3 
Gyx (Gtr), [GPa] 1.4 1.5 1.7 1.8 

Use Gxy=Gyz=Gxz (Note 2) 1.8 1.8 2.1 ~2.1 
Notes: 

1. The model Coordinate System (CS) x, y & z correspond to a Global Toroidal Coordinate 
System r (radial), t (toroidal) & p (poloidal). 

2. The transverse shear moduli are not appropriate for use in global models where beam-
bending is more characteristic than short-beam shear. An average of the in-plane shear 
values is more appropriate in all three shear directions. 

Fig. 3.0-1 is a plot of the deformed WP as a result of an imposed toroidal compression (δy) of 1 mm. 
The effective modulus is calculated by dividing the average surface stress (i.e., the total y-directed 
force divided by the y-normal area) by the toroidal strain (i.e., δy/h). Rearranging terms; 
 

E = (F h)/(Ay δy) = (25334 N x 0.094691 m)/(0.093777 m x 0.0006223 m x 0.001 m) = 41.1 GPa 
 
Where, 
 
h is the toroidal height of the winding pack, and 
Ay is the product of the radial width (w) and the poloidal thickness (dz) of the model. 
 
The Poisson’s ratio calculation requires the orthogonal (x-directed) strain in addition to the parallel 
(y-directed) strain noted above. This x-strain is simply 2δx/w. So, 
 

νxy = -εx/εy = -(2x0.0001684 m/0.093777 m)/(-0.001 m/0.094691 m) = 0.34 
 
Of course all of these calculations are incorporated into the input file using ANSYS Parametric 
Design Language (APDL) which automates the process. Calculated results are also pulled into the 
plot titles for ease of documentation, as shown in Fig. 4.0-1. 

 

Fig. 4.0-1 Toroidal Deflection Contours from Imposed Deformation 

Fig. 4.0-2 is a plot of the deformed WP from a radially-oriented loading. The loading is similar to 
the previous case. However, since there is a little less insulation in this orientation, the effective 
modulus is somewhat higher (see results in plot title). The calculations are also similar to above, 
and so they are omitted. 
 

Fig. 4.0-2 Radial Deflection Contours from Imposed Deformation 

Fig. 4.0-3 is a plot of the deformed WP from a vertically-oriented (poloidal) loading. In this case, 
the load is carried by a series of parallel springs (k=AE/L) with the Cu providing the largest 
effective stiffness. A rule of mixtures hand calculation would confirm this, although it has been 
done too many times to justify repeating it here. The ANSYS plot title captures the expected result, 
and indicates an average modulus of 83 GPa and a Poisson’s ratio close to 0.3.  
 
It is worth noting that although axial compression is applied here (even though axial tension is the 
more likely TF coil loading) the result would be the same for the tensile case. Poisson contractions 
from tensile stresses are a second or third-order effect in the stiffness calculation. 
 

 
 Fig. 4.0-3 Vertical Deflection Contours from Imposed Deformation 

Fig. 4.0-4 is a plot of the local radial deformations from a 295K-80K cooldown. An average X, Y 
and Z displacements are calculated, divided by the appropriate dimension and ΔT. Results are again 
captured in the plot title. Notice that the axial (αz) is close to but less than the value of pure Cu (14 
μ/K). The Cu value is diluted by the lower α glass-filled epoxy (in the plane of the glass weave). 
Conversely, the two in-plane contractions are dominated by the low in-plane α of the glass weave. 
 

 
Fig. 4.0-4 Deformations from Cooldown 

The average transverse shear modulus (G) is calculated in a manor similar to the elastic modulus. In 
this case the average shear stress (τ) is divided by the average shear strain (γ);  
 

G = τ/γ 
 
Fig. 4.0-5 is a plot of the deformed winding pack which is loaded by an x-directed shear relative to 
the stationary bottom plane. The total force required to achieve the 0.1 mm imposed deflection is 
divided by the shear area to get the shear stress. The imposed deflection is divided by the thickness 
of the model to get the shear strain. The numerical result is captured in the plot title. This Gzx value 
is believable because in this shear direction, the Cu carries essentially all of the stress. If the WP 
were totally Cu, then the shear modulus would come out to 51 GPa. The ratio of Cu area to total 
WP area leads to a shear modulus of 27 GPa, indicating that the insulation contributes slightly to 
this transverse modulus value. 
 
The “other” transverse shear direction Gzy is predictably similar, and therefore omitted from this 
pictorial summary. 
 

Fig. 4.0-5 Deformations from X-Directed Shear 

The in-plane shear stress presents a substantially different mechanical condition. Here, the shear 
stress must be carried almost entirely by the web of epoxy and glass, and is  
 
 Fig. 4.0-6 is a plot of the Gxy shear deformation. Again, forces are summed, turned into stress and 
divided by the average shear strain. The result is a very low shear modulus as indicated by the 1.9 
GPa value listed in the plot title. 
 
Again, the complementary Gyx shear is too similar to bother presenting. The slightly lower value is 
reported in Table 4.0-1; a value lower than this Gyx shear modulus is consistent with shearing more 
(four) layers. 

Fig. 4.0-6 Deformations from In-Plane (Gxy) Shear 

4.1 Smeared Properties Test 
 
It seems reasonable enough to assume that most of the unidirectional loadings from section 4.0 will 
produce comparable structural characteristics in a homogeneous medium with smeared orthotropic 
properties. But it is not obvious that the smeared structure will behave similar to the actual structure 
when a more complex loading is imposed (i.e., beam-like bending).  
 
A 3D model of the detailed 3x4 TF coil WP is developed as a means of benchmarking the adequacy 
of the proposed idealized model to such a loading. Fig. 4.1-1 shows a 60/2 cm length of the TF WP 
supported like a fixed-fixed beam with a uniform vertical load. Quarter-symmetry boundary 
conditions apply which cuts the beam length and width in half. The deformed structure plot shows 
that individual turns are modeled along with turn wrap and ground wrap insulation. Less visible are 
the frictionless gap elements which create a slip-plane between the conductor surface and the 49 mil 
turn insulation. In this plot, the winding pack ground wrap is 0.25” thick. However, a model with a 
3/8” GW is also developed and used. 
 

Fig. 4.1-1 Deformed Shape of Detailed TF WP “Beam” with Uniform Loading 

A homogeneous model with dimensions, boundary conditions and loading identical to those of the 
detailed WP beam is developed and shown in Fig. 4.1-2. Of course the material properties for this 
model are taken from section 4.0. The linear model runs in a flash, and shows a disconcerting result 
when compared to the nonlinear model it is expected to emulate; the deformations are low by nearly 
a factor of five. Line plots of the vertical displacement as a function of distance along the beam 
half-length are shown in Fig. 4.1-3. The blue line corresponds to the results from the smeared WP 
with orthotropic properties taken directly from Table 4.0-1 (1/4” GW). The black lines correspond 
to results from the detailed WP beam with slip-planes at the conductor surface. The red lines 
correspond to results from the smeared model with the suggested isotropic shear modulus and all 
other orthotropic properties taken from Table 4.0-1. 
 
The plot highlights the reason behind the concern which centers on the differences between shear 
deformations and bending deformations. Although the high cross-WP shear modulus calculated in 
section 4.0 (29.7 GPa) might be correct for short beam shear, it is totally inappropriate for modeling 
gross beam bending. When the lower (1.8 GPa) in-plane shear modulus is also applied to the 
transverse component, the deformations follow the detailed beam much more closely. It seems 
unlikely that any combination of material properties will make the smeared beam behave precisely 
like the detailed WP with all of its slip-planes. However, it would also seem as if adopting the lower 
in-plane shear modulus for all three shear orientations is the most reasonable compromise. 

Fig. 4.1-2 Deformed Shape of Smeared TF WP “Beam” with Uniform Loading 

This plot shows how WP “beam” deformations are determined more by gross bending than short-
beam shear. This necessitates the use of the lower in-plane shear modulus in all three shear 
directions. 
 
Notice that two different ground wrap thicknesses are presented: ¼” (solid lines) and 3/8” (dashed 
lines). Clearly, the thicker GW increases the overall bending stiffness of the WP (as indicated by a 
17% reduction in the maximum deflection). 

Fig. 4.1-3 Deformations of Detailed & Smeared TF WP Beams
From Uniform Loading with Fixed-Fixed BCs

5.0 Summary & Commentary 
 
A parametric 3D ANSYS model of the NCSX TF coil cross-section has been developed and used to 
determine equivalent smeared material properties for global models. The text-based ANSYS input 
file (Attachment 6.3) is sufficiently well annotated to allow most any knowledgeable ANSYS user 
to change dimensions, material properties, number of layers (N) and pancakes (M)…or even add 
layer or pancake insulation (presently zero), and recalculate all salient smeared properties in fairly 
short order. The resulting output must be used appropriately in a global model, taking care to 
transform this local coordinate system. The most likely transformation is presented as a footnote to 
the tabulated results (x=radial, y=toroidal, and z=poloidal).  
 
A closer review of the smeared properties and the structural characteristics of the TF coil indicate 
that the transverse shear modulus calculated by the unit-thickness model is not representative of 
bending deformations. A more accurate approach is to use the lower in-plane shear modulus in all 
three directions to capture the bending behavior of the sparsely-supported TF coil.  

6.0 Attachments 
 
6.1 Drawings 
6.2 Material Properties 
6.3 ANSYS Input File, Smeared Properties 
6.4 ANSYS Input File, Detailed WP Beam 
6.5 ANSYS Input File, Smeared WP Beam 

6.1 Drawings 

6.2 Material Properties (Taken from an earlier project memo2) 
 

Table 1.0 Material Property Data Base for use in (80K) Structural Analyses 
 

Material 

Ex 

[GPa] 

Ey 

[GPa] 

Ez 

[GPa] 

αx 

[μ/K] 

αy 

[μ/K] 

αz 

[μ/K] 

Gxy 

[GPa] 

Gxz 

[GPa] 

Gyz 

[GPa] 

υxy 

[-] 

υyz 

[-] 

υxz 

[-] 

316LN SS Annealed 209 13 81 0.283 

MCWF Casting Tensile w/SG CD w/SG  Tensile w/SG 

OFHC Cu 136 14 51 0.34 

VPI S-Glass 21 28 22 7.9 6.9 9.6 0.2 

VPI S-Glass + Kapton 

(courtesy R.P.Reed) 
19 26 21 7.9 4.2 6.0 0.2 

Prepreg S-Glass 22 30 28 7.4 5.3 (6.2) 10 0.2 

Prepreg S-Glass + Kapton 21 22 27 7.5 3.8 6.2 0.2 

VPI’d MC Conductor 37 (TBD:2x5Test) 78 CD w/SG CD Ultrasonic Twist Test  

Notes:  
1. Orthotropic material coordinate system:  

• x is the through-thickness direction 
• y is in the direction of the wrap (i.e., around the conductor perimeter) 
• z is parallel to the conductor axis 

2. Cells filled in yellow signify FE-based values. 
3. Cells filled in blue signify low-risk assumed values of comparable materials. 
4. SG = Strain Gages, CD = Cool Down 
 

                                                 
2 Leonard Myatt (Myatt Consulting, Inc.), “Material Property Data Base to be used for NCSX Analyses,” June 30, 2004. 

6.3 ANSYS Input File, Smeared Properties 
 
/batch 
rn=1 
/filnam,smear1%rn%ey 
/show,smear1%rn%,grp 
 
/prep7 
/com 
/com This input file is used to determine the smeared moduli  
/com and the coefficient of thermal expansion (CTE) of 
/com a detailed conductor pack (including GW).  
/com Look at the end of the ANSYS output listing to find the calculated  
/com  equivalent Es and alphas. 
/com 
/com mat,1 is not used 
/com mat,2 is the conductor 
/com mat,3 is the turn wrap 
/com mat,4 is layer insulation 
/com mat,5 is pancake insulation 
 
/com 
/com Run History 
/com  Run with 1e9 gap stiffness 
/com  11: 0.120" GW, all load cases 
/com  12: 0.250" GW, all load cases 
/com  13: 0.375" GW, all load cases 
/com  14: 0.500" GW, all load cases 
 
 
/com Misc Parameters 
k_nl=1                   ! invoke contact analysis 
k_dens=1                 ! integer number, 1:nominal, 2:4x more elements 
k=0.0254                 ! english to si conversion factor 
pi=acos(-1)              ! pi 
*afun,deg                ! use degrees in trig functions 
t=0.0001                 ! a tiny length 
th=0.1                   ! a tiny angle 
dtr=pi/180               ! degrees to radians 
rtd=1/dtr                ! radians to degrees 
n_lay=3                  ! number of radial layers 
n_pan=4                  ! number of pancakes 
dtmp=-215                ! differential temperature (WRT room temp) 
 
/com 
/com graphics keys 
/com 
/pnum,mat,1 
/num,1 
 
/com 
/com Element types 
/com 
et,1,82                 ! U, higher order stress element 
 
/com 
/com Material properties 
/com 

tref,295 
 
 
/com 
/com tf conductor, ignoring radii (in local CS vernacular) 
/com 
dr_con=0.966*k                        ! Conductor build in thickness (0.434*k) 
dz_con=0.709*k                        ! Conductor build in height (1.50*k) 
di_con=0.312*k                        ! inside dia of cooling channel 
 
/com 
/com Insulation 
/com 
t_tw=0.049*k              ! turn wrap insulation thk 
t_pan=0.0*k               ! pancake insulation thk (was 0.030", can be 0.0) 
t_lay=0.0*k               ! layer insulation thk (can be 0.0) 
t_gw=0.12*k               ! module over-wrap thickness (0.12" ref) 
t_gap=1e-5                ! Conductor/TW gap 
 
 
/com 
/com WP Build 
/com 
dr_wp=n_lay*(dr_con+2*t_tw)+(n_lay-1)*t_lay   ! radial build of WP 
dz_wp=n_pan*(dz_con+2*t_tw)+(n_pan-1)*t_pan   ! toroidal build of WP 
dr_iwp=dr_wp+2*t_gw                           ! radial build of ground-wrapped 
WP 
dz_iwp=dz_wp+2*t_gw                           ! toroidal build of ground-wrapped 
WP 
cel=sqrt(dr_wp**2+dz_wp**2)/k_dens            ! characteristic element size 
 
/com 
/com Cu Conductor (at 77K) 
/com 
mp,  ex,2,136e9 
mp,alpx,2,14e-6 
mp,nuxy,2,0.34 
 
/com 
/com  Turn Wrap insulation, S-glass & Kapton (X is through thickness) 
/com 
mp,  ex,3,19e9 
mp,  ey,3,26e9 
mp,  ez,3,26e9 
mp,alpx,3,21e-6 
mp,alpy,3,7.9e-6 
mp,alpz,3,7.9e-6 
mp, gxy,3,4.2e9 
mp, gyz,3,6.0e9 
mp, gxz,3,4.2e9 
mp,nuxy,3,0.2 
mp,nuyz,3,0.2 
mp,nuxz,3,0.2 
 
/com 
/com Layer to Layer insulation, S-glass (X is through thickness) 
/com 
mp,  ex,4,21e9 

mp,  ey,4,28e9 
mp,  ez,4,28e9 
mp,alpx,4,22e-6 
mp,alpy,4,7.9e-6 
mp,alpz,4,7.9e-6 
mp, gxy,4,6.9e9 
mp, gyz,4,9.6e9 
mp, gxz,4,6.9e9 
mp,nuxy,4,0.21 
mp,nuyz,4,0.21 
mp,nuxz,4,0.21 
 
/com 
/com Pancake to Pancake insulation, S-glass (local x is through thickness) 
/com 
mp,  ex,5,21e9 
mp,  ey,5,28e9 
mp,  ez,5,28e9 
mp,alpx,5,22e-6 
mp,alpy,5,7.9e-6 
mp,alpz,5,7.9e-6 
mp, gxy,5,6.9e9 
mp, gyz,5,9.6e9 
mp, gxz,5,6.9e9 
mp,nuxy,5,0.21 
mp,nuyz,5,0.21 
mp,nuxz,5,0.21 
 
/com 
/com Ground Wrap, S-glass 
/com 
mp,  ex,6,21e9 
mp,  ey,6,28e9 
mp,  ez,6,28e9 
mp,alpx,6,22e-6 
mp,alpy,6,7.9e-6 
mp,alpz,6,7.9e-6 
mp, gxy,6,6.9e9 
mp, gyz,6,9.6e9 
mp, gxz,6,6.9e9 
mp,nuxy,6,0.21 
mp,nuyz,6,0.21 
mp,nuxz,6,0.21 
 
 
/com 
/com Detailed WP 
/com 
csys 
wpcsys 
asel,none 
rectng,-dr_con/2,dr_con/2,-dz_con/2,dz_con/2 
rectng,-dr_con/2-t_tw,dr_con/2+t_tw,-dz_con/2-t_tw,dz_con/2+t_tw 
*if,t_lay,gt,0,then 
rectng,dr_con/2+t_tw,dr_con/2+t_tw+t_lay,-dz_con/2-t_tw,dz_con/2+t_tw+t_pan 
*endif 
*if,t_pan,gt,0,then 
rectng,-dr_con/2-t_tw,dr_con/2+t_tw+t_lay,dz_con/2+t_tw,dz_con/2+t_tw+t_pan 

*endif 
aovlap,all 
cm,temp,area 
/com Subtract the center hole 
pcirc,,di_con/2 
asba,temp,all 
allsel,below,area 
cm,temp,area 
/com Miter Turn Wrap Insulation 
l,kp(-dr_con/2-t_tw,-dz_con/2-t_tw,0),kp(-dr_con/2,-dz_con/2,0) 
l,kp(-dr_con/2-t_tw,+dz_con/2+t_tw,0),kp(-dr_con/2,+dz_con/2,0) 
l,kp(dr_con/2+t_tw,+dz_con/2+t_tw,0),kp(dr_con/2,+dz_con/2,0) 
l,kp(dr_con/2+t_tw,-dz_con/2-t_tw,0),kp(dr_con/2,-dz_con/2,0) 
asbl,temp,all 
cm,unit,area 
*if,k_nl,eq,1,then 
/com Make a gap between the conductor and insulation 
asel,none 
rectng,-dr_con/2-t_gap,dr_con/2+t_gap,-dz_con/2-t_gap,dz_con/2+t_gap 
cm,temp,area 
rectng,-dr_con/2,dr_con/2,-dz_con/2,dz_con/2 
asba,temp,all 
cmsel,a,unit 
asba,unit,all 
cm,unit,area 
*endif 
 
/com 
/com Overall dimensions 
/com 
cmsel,s,unit 
allsel,below,area 
csys 
*get,xmx,kp,,mxloc,x 
*get,ymx,kp,,mxloc,y 
*get,xmn,kp,,mnloc,x 
*get,ymn,kp,,mnloc,y 
dr_cell=xmx-xmn 
dz_cell=ymx-ymn 
 
/com 
/com Build array of conductors 
/com 
agen,n_lay,all,,,dr_cell 
agen,n_pan,all,,,,dz_cell 
allsel 
btol,t_gap/10 
aglue,all 
 
/com 
/com Nix extra insulation layers 
/com 
asel,all 
allsel,below,area 
*if,t_lay,gt,0,then 
*get,xmx,kp,,mxloc,x 
ksel,s,loc,x,xmx-t_lay-t,xmx+t 
lslk,,1 

asll,,1 
adele,all,,,1 
*endif 
*if,t_pan,gt,0,then 
*get,ymx,kp,,mxloc,y 
ksel,s,loc,y,ymx-t_pan-t,ymx+t 
lslk,,1 
asll,,1 
adele,all,,,1 
*endif 
 
/com 
/com Move to the center and glue on a layer of GW 
/com 
allsel 
csys 
*get,xmx,kp,,mxloc,x 
*get,ymx,kp,,mxloc,y 
*get,xmn,kp,,mnloc,x 
*get,ymn,kp,,mnloc,y 
agen,2,all,,,-(xmx+xmn)/2,-(ymx+ymn)/2,,0,,1 
*get,xmx_wp,kp,,mxloc,x 
*get,ymx_wp,kp,,mxloc,y 
*get,xmn_wp,kp,,mnloc,x 
*get,ymn_wp,kp,,mnloc,y 
 
*get,kmx,kp,,num,max 
k,kmx+1,xmx_wp+t_gw,ymx_wp+t_gw 
k,kmx+2,xmn_wp-t_gw,ymx_wp+t_gw 
k,kmx+3,xmn_wp-t_gw,ymn_wp-t_gw 
k,kmx+4,xmx_wp+t_gw,ymn_wp-t_gw 
 
a,kp(xmx_wp,ymx_wp,0),kmx+1,kmx+2,kp(xmn_wp,ymx_wp,0) 
a,kp(xmn_wp,ymx_wp,0),kmx+2,kmx+3,kp(xmn_wp,ymn_wp,0) 
a,kp(xmn_wp,ymn_wp,0),kmx+3,kmx+4,kp(xmx_wp,ymn_wp,0) 
a,kp(xmx_wp,ymn_wp,0),kmx+4,kmx+1,kp(xmx_wp,ymx_wp,0) 
aglue,all 
*get,xmx_iwp,kp,,mxloc,x 
*get,ymx_iwp,kp,,mxloc,y 
*get,xmn_iwp,kp,,mnloc,x 
*get,ymn_iwp,kp,,mnloc,y 
 
/com 
/com Make a series of local csys and set material attributes 
/com 
*do,j,1,n_pan 
*do,jj,1,n_lay 
local,100+10*j+jj,,xmn_wp+(jj-1/2)*dr_cell-t_lay/2,ymn_wp+(j-1/2)*dz_cell-
t_pan/2 
/com Conductor Attribute 
asel,s,loc,x 
asel,r,loc,y 
aatt,2 
/com Turn Wrap Attribute 
asel,s,loc,x,-dr_cell/2+t_lay,dr_cell/2-t_lay 
asel,r,loc,y,-dz_cell/2+t_pan,dz_cell/2-t_pan 
asel,u,mat,,2 
aatt,3 

/com Layer Insulation Attribute 
*if,t_lay,gt,0,then 
asel,s,loc,x,dr_cell/2-t_lay,dr_cell/2 
asel,r,loc,y,-dz_cell/2+t_pan,dz_cell/2-t_pan 
aatt,4 
*endif 
/com Pancake Insulation Attribute 
*if,t_pan,gt,0,then 
asel,s,loc,x,-dr_cell/2,dr_cell/2 
asel,r,loc,y,dz_cell/2-t_pan,dz_cell/2 
aatt,5 
*endif 
*enddo 
*enddo 
/com GW attribute 
asel,s,mat,,0 
aatt,6 
 
/com 
/com Mesh Areas 
/com 
 
/com Turn Wrap 
cel=t_tw/2 
asel,s,mat,,3 
allsel,below,area 
lsel,r,length,,,sqrt(2)*t_tw 
lesiz,all,,,2,,1 
asel,s,mat,,2,3 
allsel,below,area 
lsel,r,length,,dz_con,dz_con+2*t_tw 
lesiz,all,,,nint(dz_con/cel/2),,1 
allsel,below,area 
lsel,r,length,,dr_con,dr_con+2*t_tw 
lesiz,all,,,nint(dr_con/cel/2),,1 
asel,s,mat,,3 
allsel,below,area 
amesh,all 
 
/com Conductor 
asel,s,mat,,2 
esize,2*cel 
amesh,all 
/com Layer & Pancake Insulation 
*if,t_pan+t_lay,gt,0,then 
asel,s,mat,,4,5 
allsel,below,area 
amesh,all 
*endif 
 
/com Ground Wrap 
asel,s,mat,,6 
allsel,below,area 
lsel,r,length,,,sqrt(2)*t_gw 
nec_gw=nint(sqrt(2)*t_gw/cel/2) 
lesiz,all,,,nec_gw,,1 
allsel,below,area 
lsel,r,length,,dr_iwp-t,dr_iwp+t 

lesiz,all,,,nint(dr_iwp/cel/2),,1 
allsel,below,area 
lsel,r,length,,dz_iwp-t,dz_iwp+t 
lesiz,all,,,nint(dz_iwp/cel/2),,1 
allsel,below,area 
amesh,all 
 
/com 
/com Fix ESYS for orthotropic insulation 
/com 
local,11,,,,, 0       ! local X = global X 
local,12,,,,,90       ! local X = global Y 
/com Turn, Layer & Pancake 
*do,j,1,n_pan 
*do,jj,1,n_lay 
csys,100+10*j+jj 
asel,s,loc,y,dz_con/2,dz_con/2+t_tw+t_pan 
asel,r,loc,x,-dr_cell/2,dr_cell/2 
allsel,below,area 
emodif,all,esys,12 
asel,s,loc,y,-(dz_con/2+t_tw+t_pan),-dz_con/2 
asel,r,loc,x,-dr_cell/2,dr_cell/2 
allsel,below,area 
emodif,all,esys,12 
*enddo 
*enddo 
 
/com GW 
csys 
asel,s,loc,y,ymx_wp,ymx_iwp 
allsel,below,area 
emodif,all,esys,12 
asel,s,loc,y,ymn_iwp,ymn_wp 
allsel,below,area 
emodif,all,esys,12 
/com Turn all non-12 insulation into 11 
esel,s,mat,,3,6 
esel,u,esys,,12 
emodif,all,esys,11 
 
/com 
/com Extrude to cel long in Z 
/com 
esel,all 
nsle 
*get,e_strt,elem,,num,min 
*get,e_stop,elem,,num,max 
*get,nmx,node,,num,max 
*get,nmn,node,,num,min 
dn2=nmx-nmn+1 
csys 
ngen,2,dn2,all,,,,,cel 
/com 
/com Make the elements 
/com 
modmesh,detach 
et,11,95 
type,11 

*do,j,e_strt,e_stop,1 
*if,esel(j),ne,1,cycle 
*get,ni,elem,j,node,1 
*get,nj,elem,j,node,2 
*get,nk,elem,j,node,3 
*get,nl,elem,j,node,4 
*get,m_num,elem,j,attr,mat 
*get,e_sys,elem,j,attr,esys 
mat,m_num $esys,e_sys 
edele,j $en,j,ni,nj,nk,nl,ni+dn2,nj+dn2,nk+dn2,nl+dn2 
*enddo 
esel,all 
nsle 
nsel,invert 
ndele,all 
allsel 
emid,add 
/title,smear1%rn% Making Smeared Properties of NCSX TF WP 
aplo 
eplo 
/psym,esys,1 
/edge,1,1 
esel,u,mat,,2 
nsle 
eplo 
/psym,esys 
/edge 
 
/com 
/com Make Gap elements 
/com 
et,12,52 
r,1,1e9,0,1,,,1      ! +X 
r,2,1e9,0,1          ! +Y 
rmore,1 
r,3,1e9,0,1,,,-1     ! -X 
r,4,1e9,0,1          ! -Y 
rmore,-1 
r,5,1e9,0,1,,,1      ! +X+Y 
rmore,1 
r,6,1e9,0,1,,,-1     ! -X+Y 
rmore,1 
r,7,1e9,0,1,,,-1     ! -X-Y 
rmore,-1 
r,8,1e9,0,1,,,1      ! +X-Y 
rmore,-1 
esel,s,mat,,2,3 
nsle 
type,12 
eintf,5*t_gap,,high 
/com Fix the Gap Normals around each conductor 
*do,j,1,n_pan 
*do,jj,1,n_lay 
csys,100+10*j+jj 
esel,s,type,,12 
nsle 
nsel,r,loc,y,dz_con/2-t,dz_con/2+t_gap+t 
nsel,r,loc,x,-dr_con/2,dr_con/2 

esln,,1 
emodif,all,real,2 
esel,s,type,,12 
nsle 
nsel,r,loc,y,-dz_con/2-t_gap-t,-dz_con/2+t 
nsel,r,loc,x,-dr_con/2,dr_con/2 
esln,,1 
emodif,all,real,4 
esel,s,type,,12 
nsle 
nsel,r,loc,x,-dr_con/2-t_gap-t,-dr_con/2+t 
nsel,r,loc,y,-dz_con/2,dz_con/2 
esln,,1 
emodif,all,real,3 
/com Fix the corner gaps 
esel,s,type,,12 
nsle 
nsel,r,loc,y,dz_con/2-t,dz_con/2+t_gap+t 
nsel,r,loc,x,dr_con/2-t,dr_con/2+t_gap+t 
esln,,1 
emodif,all,real,5 
esel,s,type,,12 
nsle 
nsel,r,loc,y,-dz_con/2-t_gap-t,-dz_con/2+t 
nsel,r,loc,x,-dr_con/2-t_gap-t,-dr_con/2+t 
esln,,1 
emodif,all,real,7 
esel,s,type,,12 
nsle 
nsel,r,loc,x,-dr_con/2-t_gap-t,-dr_con/2+t 
nsel,r,loc,y,dz_con/2-t,dz_con/2+t_gap+t 
esln,,1 
emodif,all,real,6 
esel,s,type,,12 
nsle 
nsel,r,loc,x,dr_con/2-t,dr_con/2+t_gap+t 
nsel,r,loc,y,-dz_con/2-t_gap-t,-dz_con/2+t 
esln,,1 
emodif,all,real,8 
*enddo 
*enddo 
 
/com WP Dims 
allsel 
csys 
*get,xmn,node,,mnloc,x 
*get,xmx,node,,mxloc,x 
*get,ymn,node,,mnloc,y 
*get,ymx,node,,mxloc,y 
*get,zmx,node,,mxloc,z 
*get,zmn,node,,mnloc,z 
 
fini 
 
/solu 
allsel 
bfunif,temp,295 
/com 

/com Ey BCs 
/com 
nsel,s,loc,y,ymn-t,ymn+t 
d,all,uy 
cm,y_grnd,node 
nsel,s,loc,y,ymx-t,ymx+t 
cm,y_forc,node 
d,all,uy,-0.001 
nsel,s,loc,z 
d,all,uz 
esel,all 
nsle 
nsel,r,loc,z,zmx 
cp,next,uz,all 
nsel,s,loc,x 
d,all,ux 
allsel 
!autots,on 
!nsubst,10,1000,2 
solve 
fini 
/com 
/com Ey Postprocessing 
/com 
/post1 
allsel 
cmsel,s,y_forc 
fsum 
*get,fy,fsum,,item,fy 
ay=(xmx-xmn)*(zmx-zmn) 
csys 
*get,u_y,node,node(xmx,ymx,0),u,y 
yy_strain=u_y/(ymx-ymn) 
ey=abs(0.1*nint((1e-8)*(fy/ay)/yy_strain)) 
nsel,s,loc,x,xmx 
esln 
nsle 
etab,ux,u,x 
etab,volu,volu 
smult,uxv,ux,volu 
ssum 
*get,uxvtot,ssum,,item,uxv 
*get,vtot,ssum,,item,volu 
u_x=uxvtot/vtot 
xy_strain=u_x/(xmx) 
nu_xy=0.001*nint(-1000*xy_strain/yy_strain) 
/title,smear1%rn% SA of NCSX TF WP, Ey=%ey% GPa, NUxy=%nu_xy% 
allsel 
plns,u,y 
save 
fini 
 
/filnam,smear1%rn%ex 
/solu 
/com 
/com Ex BCs 
/com 
allsel 

ddele,all,uy 
ddele,all,ux 
nsel,s,loc,x,xmn-t,xmn+t 
d,all,ux 
cm,x_grnd,node 
nsel,s,loc,x,xmx-t,xmx+t 
cm,x_forc,node 
d,all,ux,-0.001 
nsel,s,loc,y 
d,all,uy 
allsel 
!autots,on 
!nsubst,10,1000,2 
solve 
fini 
 
/com 
/com Ex Postprocessing 
/com 
/post1 
allsel 
cmsel,s,x_forc 
fsum 
*get,fx,fsum,,item,fx 
ax=(ymx-ymn)*(zmx-zmn) 
csys 
*get,u_x,node,node(xmx,ymx,0),u,x 
xx_strain=u_x/(xmx-xmn) 
ex=abs(0.1*nint((1e-8)*(fx/ax)/xx_strain)) 
nsel,s,loc,y,ymx 
esln 
nsle 
etab,uy,u,y 
etab,volu,volu 
smult,uyv,uy,volu 
ssum 
*get,uyvtot,ssum,,item,uyv 
*get,vtot,ssum,,item,volu 
u_y=uyvtot/vtot 
yx_strain=u_y/(ymx) 
nu_yx=0.001*nint(-1000*yx_strain/xx_strain) 
/title,smear1%rn% SA of NCSX TF WP, Ex=%ex% GPa, NUyx=%nu_yx% 
allsel 
plns,u,y 
save 
fini 
 
 
/filnam,smear1%rn%ez 
/solu 
/com 
/com Ez BCs 
/com 
allsel 
ddele,all,uy 
ddele,all,ux 
nsel,s,loc,x 
d,all,ux 

nsel,s,loc,y 
d,all,uy 
cpdele,all 
nsel,s,loc,z 
cm,z_grnd,node 
nsel,s,loc,z,zmx-t,zmx+t 
cm,z_forc,node 
d,all,uz,-0.00001 
allsel 
solve 
fini 
/com 
/com Ez Postprocessing 
/com 
/post1 
allsel 
cmsel,s,z_forc 
fsum 
*get,fz,fsum,,item,fz 
az=(ymx-ymn)*(xmx-xmn) 
csys 
*get,u_z,node,node(xmx,ymx,zmx),u,z 
zz_strain=u_z/(zmx-zmn) 
ez=abs(0.1*nint((1e-8)*(fz/az)/zz_strain)) 
nsel,s,loc,y,ymx 
esln 
nsle 
etab,uy,u,y 
etab,volu,volu 
smult,uyv,uy,volu 
ssum 
*get,uyvtot,ssum,,item,uyv 
*get,vtot,ssum,,item,volu 
u_y=uyvtot/vtot 
yz_strain=u_y/(ymx) 
nsel,s,loc,x,xmx 
esln 
nsle 
etab,ux,u,x 
etab,volu,volu 
smult,uxv,ux,volu 
ssum 
*get,uxvtot,ssum,,item,uxv 
*get,vtot,ssum,,item,volu 
u_x=uxvtot/vtot 
xz_strain=u_x/(xmx) 
ave_strain=(yz_strain+xz_strain)/2 
 
nu_xyz=0.001*nint(-1000*ave_strain/zz_strain) 
/title,smear1%rn% SA of NCSX TF WP, Ez=%ez% GPa, NUxyz=%nu_xyz% 
allsel 
plns,u,z 
save 
fini 
 
 
/filnam,smear1%rn%alpha 
/solu 

/com 
/com Alpha BCs 
/com 
allsel 
cpdele,all 
nsel,s,loc,z,zmx 
ddele,all,uz 
cp,next,uz,all 
allsel 
bfunif,temp,80 
solve 
fini 
/com 
/com Alpha Postprocessing 
/com 
/post1 
allsel 
csys 
*get,u_z,node,node(xmx,ymx,zmx),u,z 
z_strain=u_z/(zmx-zmn) 
alphaz=abs(0.1*nint((1e7)*z_strain/215)) 
nsel,s,loc,y,ymx 
esln 
nsle 
etab,uy,u,y 
etab,volu,volu 
smult,uyv,uy,volu 
ssum 
*get,uyvtot,ssum,,item,uyv 
*get,vtot,ssum,,item,volu 
u_y=uyvtot/vtot 
y_strain=u_y/ymx 
alphay=abs(0.1*nint((1e7)*y_strain/215)) 
 
nsel,s,loc,x,xmx 
esln 
nsle 
etab,ux,u,x 
etab,volu,volu 
smult,uxv,ux,volu 
ssum 
*get,uxvtot,ssum,,item,uxv 
*get,vtot,ssum,,item,volu 
u_x=uxvtot/vtot 
x_strain=u_x/xmx 
alphax=abs(0.1*nint((1e7)*x_strain/215)) 
 
/title,smear1%rn% SA of NCSX TF WP, AlphaX/Y/Z=%alphax%/%alphay%/%alphaz% mico/K 
allsel 
plns,u,x 
plns,u,y 
plns,u,z 
save 
fini 
 
 
/filnam,smear1%rn%tauzx 
/solu 

/com 
/com TauZX BCs 
/com 
allsel 
ddele,all,ux 
nsel,s,loc,y 
d,all,uy 
nsel,s,loc,z,zmx 
d,all,ux,0.00001 
nsel,s,loc,z 
d,all,ux 
allsel 
bfunif,temp,295 
solve 
fini 
 
/com 
/com TauZX Postprocessing 
/com 
/post1 
allsel 
csys 
nsel,s,loc,z 
fsum 
*get,fx,fsum,,item,fx 
tauzx=fx/az 
allsel 
*get,u_zx,node,node(xmx,ymx,zmx),u,x 
gammazx=u_zx/(zmx-zmn) 
gzx=0.1*nint(1e-8*tauzx/gammazx) 
/title,smear1%rn% SA of NCSX TF WP, TAUzx=%gzx% GPa 
plns,u,x 
save 
fini 
 
 
/filnam,smear1%rn%tauzy 
/solu 
/com 
/com TAUzy BCs 
/com 
allsel 
ddele,all,uy 
ddele,all,ux 
nsel,s,loc,x 
d,all,ux 
nsel,s,loc,z,zmx 
d,all,uy,0.00001 
nsel,s,loc,z 
d,all,uy 
allsel 
bfunif,temp,295 
solve 
fini 
 
/com 
/com TAUzy Postprocessing 
/com 

/post1 
allsel 
csys 
nsel,s,loc,z 
fsum 
*get,fy,fsum,,item,fy 
tauzy=fy/az 
allsel 
*get,u_zy,node,node(xmx,ymx,zmx),u,y 
gammazy=u_zy/(zmx-zmn) 
gzy=0.1*nint(1e-8*tauzy/gammazy) 
/title,smear1%rn% SA of NCSX TF WP, TAUzy=%gzy% GPa 
plns,u,y 
save 
fini 
 
 
/filnam,smear1%rn%tauxy 
/solu 
/com 
/com TAUxy BCs 
/com 
allsel 
ddele,all,uy 
ddele,all,ux 
ddele,all,uz 
nsel,s,loc,x,xmn-t,xmn+t 
cm,n_grnd,node 
d,all,uy 
d,all,ux 
nsel,s,loc,x,xmx-t,xmx+t 
d,all,uy,0.001 
d,all,ux 
nsel,s,loc,z,zmn 
d,all,uz 
nsel,s,loc,x 
nsel,r,loc,y 
d,all,ux 
 
allsel 
bfunif,temp,295 
solve 
fini 
 
/com 
/com TAUxy Postprocessing 
/com 
/post1 
allsel 
csys 
cmsel,s,n_grnd 
fsum 
*get,fy,fsum,,item,fy 
tauxy=fy/ax 
allsel 
*get,u_xy,node,node(xmx,ymx,zmx),u,y 
gammaxy=u_xy/(xmx-xmn) 
gxy=0.1*nint(1e-8*tauxy/gammaxy) 

/title,smear1%rn% SA of NCSX TF WP, TAUxy=%gxy% GPa 
plns,u,y 
save 
fini 
 
 
 
/filnam,smear1%rn%tauyx 
/solu 
/com 
/com TAUyx BCs 
/com 
allsel 
ddele,all,uy 
ddele,all,ux 
ddele,all,uz 
nsel,s,loc,y,ymn-t,ymn+t 
cm,n_grnd,node 
d,all,uy 
d,all,ux 
nsel,s,loc,y,ymx-t,ymx+t 
d,all,ux,0.001 
d,all,uy 
nsel,s,loc,z,zmn 
d,all,uz 
nsel,s,loc,x 
nsel,r,loc,y 
d,all,uy 
 
allsel 
bfunif,temp,295 
solve 
fini 
 
/com 
/com TAUyx Postprocessing 
/com 
/post1 
allsel 
csys 
cmsel,s,n_grnd 
fsum 
*get,fx,fsum,,item,fx 
tauyx=fx/ay 
allsel 
*get,u_yx,node,node(xmx,ymx,zmx),u,x 
gammayx=u_yx/(ymx-ymn) 
gyx=0.1*nint(1e-8*tauyx/gammayx) 
/title,smear1%rn% SA of NCSX TF WP, TAUyx=%gyx% GPa 
plns,u,x 
save 
fini 
 
/exit,nosa 
/eof 

6.4 ANSYS Input File, Detailed WP Beam 
 
/batch 
rn=1 
/filnam,wpbeam2%rn% 
/show,wpbeam2%rn%,grp 
!resume 
!*if,1,eq,1,:1000 
/prep7 
/com 
/com Structural model of 3x4 NCSX TF WP as a beam for stiffness calc with a 
slip-plane. 
/com 
/com Run History 
/com 
/com 20: unbonded, 1/4" GW, Uniform Loading on Fixed-Fixed beam (questionable 
gap behavior) 
/com 21: adopt contact elements from smear1.dat 
 
/com 
/com Misc Parameters 
/com 
load_dir=1               ! 1: X, 2: Y 
em_pres=1e6              ! magnetic pressure loading 
k_nl=1                   ! 0: Bonded, 1: N-N contact 
k_dens=1                 ! 1: nominal mesh size, >1 smaller mesh size, <1 bigger 
mesh size 
l_tot=0.30               ! beam length 
mu_cont=0.0              ! Contact friction coeffient 
k=0.0254                 ! english to si conversion factor 
pi=acos(-1)              ! pi 
mu0=4*pi*1e-7            ! Mu0 
*afun,deg                ! use degrees in trig functions 
t=0.0001                 ! a tiny length 
n_lay=3                  ! number of radial layers 
n_pan=4                  ! number of pancakes 
dtmp=0!-215                ! differential temperature (WRT room temp) 
 
/com 
/com element types 
/com 
et,1,45                  ! structure only 
et,2,42                  ! planar element 
et,12,52                 ! gap elements 
r,1,1e9,0,1,,,1      ! +X 
r,2,1e9,0,1          ! +Y 
rmore,1 
r,3,1e9,0,1,,,-1     ! -X 
r,4,1e9,0,1          ! -Y 
rmore,-1 
r,5,1e9,0,1,,,1      ! +X+Y 
rmore,1 
r,6,1e9,0,1,,,-1     ! -X+Y 
rmore,1 
r,7,1e9,0,1,,,-1     ! -X-Y 
rmore,-1 
r,8,1e9,0,1,,,1      ! +X-Y 
rmore,-1 

/com 
/com graphics keys 
/com 
/pnum,mat,1 
/num,1 
/dist 
/focus 
/vup,1,x 
/view,1,,-1 
 
/com 
/com tf conductor, ignoring radii (in local CS vernacular) 
/com 
dr_con=0.966*k                       ! Conductor build in thickness 
dz_con=0.709*k                       ! Conductor build in height 
di_con=0.312*k                       ! inside diameter of cooling channel 
 
/com 
/com Insulation 
/com 
t_tw=0.049*k              ! turn wrap insulation thk 
t_pan=0.0*k               ! pancake insulation thk (can be 0.0) 
t_lay=0.0*k               ! layer insulation thk (can be 0.0) 
t_gw=0.25*k               ! module over-wrap thickness 
t_gap=1e-5                ! Conductor/TW gap 
 
/com 
/com WP Build 
/com 
dr_wp=n_lay*(dr_con+2*t_tw)+(n_lay-1)*t_lay   ! radial build of WP 
dz_wp=n_pan*(dz_con+2*t_tw)+(n_pan-1)*t_pan   ! toroidal build of WP 
dr_iwp=dr_wp+2*t_gw                           ! radial build of ground-wrapped 
WP 
dz_iwp=dz_wp+2*t_gw                           ! toroidal build of ground-wrapped 
WP 
cel=sqrt(dr_wp**2+dz_wp**2)/k_dens            ! characteristic element size 
 
 
/com 
/com material properties 
/com 
/com 
/com Cu Conductor (at 77K) 
/com 
mp,  ex,2,136e9 
mp,alpx,2,14e-6 
mp,nuxy,2,0.34 
 
/com 
/com  Turn Wrap insulation, S-glass & Kapton (X is through thickness) 
/com 
mp,  ex,3,19e9 
mp,  ey,3,26e9 
mp,  ez,3,26e9 
mp,alpx,3,21e-6 
mp,alpy,3,7.9e-6 
mp,alpz,3,7.9e-6 

mp, gxy,3,4.2e9 
mp, gyz,3,6.0e9 
mp, gxz,3,4.2e9 
mp,nuxy,3,0.2 
mp,nuyz,3,0.2 
mp,nuxz,3,0.2 
 
/com 
/com Layer to Layer insulation, S-glass (X is through thickness) 
/com 
mp,  ex,4,21e9 
mp,  ey,4,28e9 
mp,  ez,4,28e9 
mp,alpx,4,22e-6 
mp,alpy,4,7.9e-6 
mp,alpz,4,7.9e-6 
mp, gxy,4,6.9e9 
mp, gyz,4,9.6e9 
mp, gxz,4,6.9e9 
mp,nuxy,4,0.21 
mp,nuyz,4,0.21 
mp,nuxz,4,0.21 
 
/com 
/com Pancake to Pancake insulation, S-glass (local x is through thickness) 
/com 
mp,  ex,5,21e9 
mp,  ey,5,28e9 
mp,  ez,5,28e9 
mp,alpx,5,22e-6 
mp,alpy,5,7.9e-6 
mp,alpz,5,7.9e-6 
mp, gxy,5,6.9e9 
mp, gyz,5,9.6e9 
mp, gxz,5,6.9e9 
mp,nuxy,5,0.21 
mp,nuyz,5,0.21 
mp,nuxz,5,0.21 
 
/com 
/com Ground Wrap, S-glass 
/com 
mp,  ex,6,21e9 
mp,  ey,6,28e9 
mp,  ez,6,28e9 
mp,alpx,6,22e-6 
mp,alpy,6,7.9e-6 
mp,alpz,6,7.9e-6 
mp, gxy,6,6.9e9 
mp, gyz,6,9.6e9 
mp, gxz,6,6.9e9 
mp,nuxy,6,0.21 
mp,nuyz,6,0.21 
mp,nuxz,6,0.21 
 
 
/com 
/com Gaps 

/com 
mp,mu,10,mu_cont 
 
/com 
/com Detailed WP 
/com 
csys 
wpcsys 
asel,none 
rectng,-dr_con/2,dr_con/2,-dz_con/2,dz_con/2 
rectng,-dr_con/2-t_tw,dr_con/2+t_tw,-dz_con/2-t_tw,dz_con/2+t_tw 
*if,t_lay,gt,0,then 
rectng,dr_con/2+t_tw,dr_con/2+t_tw+t_lay,-dz_con/2-t_tw,dz_con/2+t_tw+t_pan 
*endif 
*if,t_pan,gt,0,then 
rectng,-dr_con/2-t_tw,dr_con/2+t_tw+t_lay,dz_con/2+t_tw,dz_con/2+t_tw+t_pan 
*endif 
aovlap,all 
cm,temp,area 
/com Subtract the center hole 
pcirc,,di_con/2 
asba,temp,all 
allsel,below,area 
cm,temp,area 
/com Miter Turn Wrap Insulation 
l,kp(-dr_con/2-t_tw,-dz_con/2-t_tw,0),kp(-dr_con/2,-dz_con/2,0) 
l,kp(-dr_con/2-t_tw,+dz_con/2+t_tw,0),kp(-dr_con/2,+dz_con/2,0) 
l,kp(dr_con/2+t_tw,+dz_con/2+t_tw,0),kp(dr_con/2,+dz_con/2,0) 
l,kp(dr_con/2+t_tw,-dz_con/2-t_tw,0),kp(dr_con/2,-dz_con/2,0) 
asbl,temp,all 
cm,unit,area 
*if,k_nl,eq,1,then 
/com Make a gap between the conductor and insulation 
asel,none 
rectng,-dr_con/2-t_gap,dr_con/2+t_gap,-dz_con/2-t_gap,dz_con/2+t_gap 
cm,temp,area 
rectng,-dr_con/2,dr_con/2,-dz_con/2,dz_con/2 
asba,temp,all 
cmsel,a,unit 
asba,unit,all 
cm,unit,area 
*endif 
 
/com 
/com Overall dimensions 
/com 
cmsel,s,unit 
allsel,below,area 
csys 
*get,xmx,kp,,mxloc,x 
*get,ymx,kp,,mxloc,y 
*get,xmn,kp,,mnloc,x 
*get,ymn,kp,,mnloc,y 
dr_cell=xmx-xmn 
dz_cell=ymx-ymn 
 
/com 
/com Build array of conductors 

/com 
agen,n_lay,all,,,dr_cell 
agen,n_pan,all,,,,dz_cell 
allsel 
btol,t_gap/10 
aglue,all 
 
/com 
/com Nix extra insulation layers 
/com 
asel,all 
allsel,below,area 
*if,t_lay,gt,0,then 
*get,xmx,kp,,mxloc,x 
ksel,s,loc,x,xmx-t_lay-t,xmx+t 
lslk,,1 
asll,,1 
adele,all,,,1 
*endif 
*if,t_pan,gt,0,then 
*get,ymx,kp,,mxloc,y 
ksel,s,loc,y,ymx-t_pan-t,ymx+t 
lslk,,1 
asll,,1 
adele,all,,,1 
*endif 
 
/com 
/com Move to the center and glue on a layer of GW 
/com 
allsel 
csys 
*get,xmx,kp,,mxloc,x 
*get,ymx,kp,,mxloc,y 
*get,xmn,kp,,mnloc,x 
*get,ymn,kp,,mnloc,y 
agen,2,all,,,-(xmx+xmn)/2,-(ymx+ymn)/2,,0,,1 
*get,xmx_wp,kp,,mxloc,x 
*get,ymx_wp,kp,,mxloc,y 
*get,xmn_wp,kp,,mnloc,x 
*get,ymn_wp,kp,,mnloc,y 
 
*get,kmx,kp,,num,max 
k,kmx+1,xmx_wp+t_gw,ymx_wp+t_gw 
k,kmx+2,xmn_wp-t_gw,ymx_wp+t_gw 
k,kmx+3,xmn_wp-t_gw,ymn_wp-t_gw 
k,kmx+4,xmx_wp+t_gw,ymn_wp-t_gw 
 
a,kp(xmx_wp,ymx_wp,0),kmx+1,kmx+2,kp(xmn_wp,ymx_wp,0) 
a,kp(xmn_wp,ymx_wp,0),kmx+2,kmx+3,kp(xmn_wp,ymn_wp,0) 
a,kp(xmn_wp,ymn_wp,0),kmx+3,kmx+4,kp(xmx_wp,ymn_wp,0) 
a,kp(xmx_wp,ymn_wp,0),kmx+4,kmx+1,kp(xmx_wp,ymx_wp,0) 
aglue,all 
*get,xmx_iwp,kp,,mxloc,x 
*get,ymx_iwp,kp,,mxloc,y 
*get,xmn_iwp,kp,,mnloc,x 
*get,ymn_iwp,kp,,mnloc,y 

/com 
/com Overlay Lines at Symmetry Planes 
/com 
allsel 
cm,wp,area 
*get,kmx,kp,,num,max 
csys 
k,kmx+1,-dr_iwp/2 
k,kmx+2,+dr_iwp/2 
l,kmx+1,kmx+2 
k,kmx+3,,-dz_iwp/2 
k,kmx+4,,+dz_iwp/2 
!l,kmx+3,kmx+4     ! forget about symmetry this way 
asbl,wp,all 
 
/com 
/com Make a series of local csys and set material attributes 
/com 
*do,j,1,n_pan 
*do,jj,1,n_lay 
local,100+10*j+jj,,xmn_wp+(jj-1/2)*dr_cell-t_lay/2,ymn_wp+(j-1/2)*dz_cell-
t_pan/2 
/com Conductor Attribute 
asel,s,loc,x 
asel,r,loc,y 
aatt,2 
/com Turn Wrap Attribute 
asel,s,loc,x,-dr_cell/2+t_lay,dr_cell/2-t_lay 
asel,r,loc,y,-dz_cell/2+t_pan,dz_cell/2-t_pan 
asel,u,mat,,2 
aatt,3 
/com Layer Insulation Attribute 
*if,t_lay,gt,0,then 
asel,s,loc,x,dr_cell/2-t_lay,dr_cell/2 
asel,r,loc,y,-dz_cell/2+t_pan,dz_cell/2-t_pan 
aatt,4 
*endif 
/com Pancake Insulation Attribute 
*if,t_pan,gt,0,then 
asel,s,loc,x,-dr_cell/2,dr_cell/2 
asel,r,loc,y,dz_cell/2-t_pan,dz_cell/2 
aatt,5 
*endif 
*enddo 
*enddo 
/com GW attribute 
asel,s,mat,,0 
aatt,6 
 
/com 
/com Mesh Areas 
/com 
 
/com Turn Wrap 
cel=sqrt(2)*t_tw 
asel,s,mat,,3 
allsel,below,area 
lsel,r,length,,,sqrt(2)*t_tw 

lesiz,all,,,1,,1 
asel,s,mat,,2,3 
allsel,below,area 
lsel,r,length,,dz_con,dz_con+2*t_tw 
lesiz,all,,,nint(dz_con/cel/2),,1 
allsel,below,area 
lsel,r,length,,dr_con,dr_con+2*t_tw 
lesiz,all,,,nint(dr_con/cel/2),,1 
asel,s,mat,,3 
allsel,below,area 
amesh,all 
 
/com Conductor 
asel,s,mat,,2 
esize,2*cel 
amesh,all 
/com Layer & Pancake Insulation 
*if,t_pan+t_lay,gt,0,then 
asel,s,mat,,4,5 
allsel,below,area 
amesh,all 
*endif 
 
/com Ground Wrap 
asel,s,mat,,6 
allsel,below,area 
lsel,r,length,,,sqrt(2)*t_gw 
nec_gw=nint(sqrt(2)*t_gw/cel/2) 
lesiz,all,,,nec_gw,,1 
allsel,below,area 
lsel,r,length,,dr_iwp-t,dr_iwp+t 
lesiz,all,,,nint(dr_iwp/cel/2),,1 
allsel,below,area 
lsel,r,length,,dz_iwp-t,dz_iwp+t 
lesiz,all,,,nint(dz_iwp/cel/2),,1 
allsel,below,area 
amesh,all 
 
/com 
/com Fix ESYS for orthotropic insulation 
/com 
local,11,,,,, 0       ! local X = global X 
local,12,,,,,90       ! local X = global Y 
/com Turn, Layer & Pancake 
*do,j,1,n_pan 
*do,jj,1,n_lay 
csys,100+10*j+jj 
asel,s,loc,y,dz_con/2,dz_con/2+t_tw+t_pan 
asel,r,loc,x,-dr_cell/2,dr_cell/2 
allsel,below,area 
emodif,all,esys,12 
asel,s,loc,y,-(dz_con/2+t_tw+t_pan),-dz_con/2 
asel,r,loc,x,-dr_cell/2,dr_cell/2 
allsel,below,area 
emodif,all,esys,12 
*enddo 
*enddo 

/com GW 
csys 
asel,s,loc,y,ymx_wp,ymx_iwp 
allsel,below,area 
emodif,all,esys,12 
asel,s,loc,y,ymn_iwp,ymn_wp 
allsel,below,area 
emodif,all,esys,12 
/com Turn all non-12 insulation into 11 
esel,s,mat,,3,6 
esel,u,esys,,12 
emodif,all,esys,11 
 
 
/com 
/com Extrude in Z 
/com 
allsel 
modmesh,detach 
*get,e_strt,elem,,num,min 
*get,e_stop,elem,,num,max 
*get,nmx,node,,num,max 
*get,nmn,node,,num,min 
dn2=nmx-nmn+1 
nez=nint(l_tot/cel) 
nez=nint(l_tot/cel/3) 
 
/com 
/com Generate nodes 
/com 
allsel 
ngen,nez+1,dn2,all,,,,,l_tot/nez 
 
/com 
/com Make the elements 
/com 
modmesh,detach 
type,1 
*do,j,e_strt,e_stop,1 
*if,esel(j),ne,1,cycle 
*get,ni,elem,j,node,1 
*get,nj,elem,j,node,2 
*get,nk,elem,j,node,3 
*get,nl,elem,j,node,4 
*get,m_num,elem,j,attr,mat 
*get,esys_num,elem,j,attr,esys 
mat,m_num $esys,esys_num $edele,j $en,j,ni,nj,nk,nl,ni+dn2,nj+dn2,nk+dn2,nl+dn2 
*enddo 
egen,nez,dn2,all 
 
/com 
/com Deal with the conductor/insulator interfaces 
/com 
*if,k_nl,eq,0,then 
allsel 
numm,node 
*else 
/com Create Contact Interface (i-node on conductor, j-node on insulation) 



43 

allsel 
mat,10 $type,12 $real,1 
eintf,2*t_gap,,high 
 
/com Fix the Gap Normals around each conductor 
*do,j,1,n_pan 
*do,jj,1,n_lay 
csys,100+10*j+jj 
esel,s,type,,12 
nsle 
nsel,r,loc,y,dz_con/2-t,dz_con/2+t_gap+t 
nsel,r,loc,x,-dr_con/2,dr_con/2 
esln,,1 
emodif,all,real,2 
esel,s,type,,12 
nsle 
nsel,r,loc,y,-dz_con/2-t_gap-t,-dz_con/2+t 
nsel,r,loc,x,-dr_con/2,dr_con/2 
esln,,1 
emodif,all,real,4 
esel,s,type,,12 
nsle 
nsel,r,loc,x,-dr_con/2-t_gap-t,-dr_con/2+t 
nsel,r,loc,y,-dz_con/2,dz_con/2 
esln,,1 
emodif,all,real,3 
/com Fix the corner gaps 
esel,s,type,,12 
nsle 
nsel,r,loc,y,dz_con/2-t,dz_con/2+t_gap+t 
nsel,r,loc,x,dr_con/2-t,dr_con/2+t_gap+t 
esln,,1 
emodif,all,real,5 
esel,s,type,,12 
nsle 
nsel,r,loc,y,-dz_con/2-t_gap-t,-dz_con/2+t 
nsel,r,loc,x,-dr_con/2-t_gap-t,-dr_con/2+t 
esln,,1 
emodif,all,real,7 
esel,s,type,,12 
nsle 
nsel,r,loc,x,-dr_con/2-t_gap-t,-dr_con/2+t 
nsel,r,loc,y,dz_con/2-t,dz_con/2+t_gap+t 
esln,,1 
emodif,all,real,6 
esel,s,type,,12 
nsle 
nsel,r,loc,x,dr_con/2-t,dr_con/2+t_gap+t 
nsel,r,loc,y,-dz_con/2-t_gap-t,-dz_con/2+t 
esln,,1 
emodif,all,real,8 
*enddo 
*enddo 
 
/com 
/com Make a symmetric model 
/com 
csys 

*if,load_dir,eq,1,then 
nsel,s,loc,y,-12,t 
esln,,1 
edele,all 
esel,all 
nsle 
nsel,invert 
ndele,all 
nsel,s,loc,y 
d,all,uy 
*elseif,load_dir,eq,2 
nsel,s,loc,x,-12,t 
esln,,1 
edele,all 
esel,all 
nsle 
nsel,invert 
ndele,all 
nsel,s,loc,x 
d,all,ux 
*endif 
 
/com 
/com BCs 
/com 
csys 
nsel,s,loc,z 
d,all,uz 
allsel 
nsel,s,loc,z,l_tot 
cp,next,uz,all 
 
/com 
/com Apply a Load & Restrain Load Direction 
csys 
*if,load_dir,eq,1,then 
esel,all 
nsel,s,loc,x,dr_iwp/2 
sf,all,pres,em_pres 
nsel,s,loc,x,-dr_iwp/2 
nsel,r,loc,z,,l_tot/nez+t 
d,all,ux 
*elseif,load_dir,eq,2 
esel,all 
nsel,s,loc,y,dz_iwp/2 
sf,all,pres,em_pres 
nsel,s,loc,y,-dz_iwp/2 
nsel,r,loc,z,,l_tot/nez+t 
d,all,uy 
*endif 
 
/title,wpbeam2%rn%, Bending Def of TFWP 
allsel 
eplo 
allsel 
save 
fini 
:100 

/solu 
tref,0 
bfunif,temp,dtmp 
*if,k_nl,ne,0,then 
autots,on 
nsubst,2000,4000,2 
nropt,auto 
*endif 
solve 
fini 
 
:1000 
/auto 
/post1 
 
/auto 
/view,1,1,1,1 
/title,wpbeam2%rn%, Deformed Shape 
esel,s,mat,,2 
nsle 
plns,s,int 
 
/com 
/com WP Insulation Stress Evaluation 
/com 
/com constants for TPX Epoxy/S-2 glass Insulation 
/com 
*dim,insul,,6       ! 
GW(Normal),GW(Shear),GW(Norm+Shear),WP(Normal),WP(Shear),WP(Norm+Shear) 
rsys,solu 
bs=40e6             ! Bond Strength (t0) 
cf=0.51             ! Friction Coefficient (c2) 
sn_small=4.4e6      ! threshold for allowing small tensile stress (0.02% strain 
in 22 GPa Insul) 
/com GW 
esel,s,mat,,6 
cm,gw,elem 
etable,volgw,volu 
ssum 
*get,vgw,ssum,,item,volgw 
etable,sx,s,x 
etable,sy,s,y 
etable,sz,s,z 
etable,sxy,s,xy 
etable,sxz,s,xz 
smult,sxy2,sxy,sxy 
smult,sxz2,sxz,sxz 
sadd,sxy2xz2,sxy2,sxz2 
sexp,sxyxz,sxy2xz2,,0.5 
 
/com WP Insulation 
esel,s,mat,,3,5 
cm,wp,elem 
etable,volwp,volu 
ssum 
*get,vwp,ssum,,item,volwp 
esel,s,mat,,3,4 
etable,sx,s,x 

etable,sy,s,y 
etable,sz,s,z 
esel,s,mat,,5 
etable,sx,s,z 
etable,sy,s,y 
etable,sz,s,x 
esel,s,mat,,3,5 
etable,sxy,s,xy 
etable,sxz,s,xz 
smult,sxy2,sxy,sxy 
smult,sxz2,sxz,sxz 
sadd,sxy2xz2,sxy2,sxz2 
sexp,sxyxz,sxy2xz2,,0.5 
 
/com 
/com Process GWI 
/com 
cmsel,s,gw 
/com Shear Allowable based on bs and cf 
sadd,shallgw,sx,,-cf,,2*bs/3 
/com Shear Margin = Shear Allowable - Local Shear 
sadd,smgw,shallgw,sxyxz,1,-1 
/com Elements with local shear margin 
esel,r,etab,smgw,.001,1e12 
cm,oktgw,elem 
ssum 
*get,voktgw,ssum,,item,volgw 
/com Elements with positive shear margin, and Local normal stress less than 
Limit 
cmsel,s,gw 
esel,r,etab,sx,-1e12,sn_small 
cm,okngw,elem 
ssum 
*get,vokngw,ssum,,item,volgw 
cmsel,s,oktgw 
esel,r,etab,sx,-1e12,sn_small 
cm,oktngw,elem 
ssum 
*get,voktngw,ssum,,item,volgw 
 
/com 
/com Process WPI 
/com 
cmsel,s,wp 
/com Shear Allowable based on bs and cf 
sadd,shallwp,sx,,-cf,,2*bs/3 
/com Shear Margin = Shear Allowable - Local Shear 
sadd,smwp,shallwp,sxyxz,1,-1 
/com Elements with positive shear margin 
esel,r,etab,smwp,.001,1e12 
cm,oktwp,elem 
ssum 
*get,voktwp,ssum,,item,volwp 
/com WPI Elements with positive shear margin, and Negative normal 
cmsel,s,wp 
esel,r,etab,sx,-1e12,sn_small 
cm,oknwp,elem 
ssum 

*get,voknwp,ssum,,item,volwp 
cmsel,s,oktwp 
esel,r,etab,sx,-1e12,sn_small 
cm,oktnwp,elem 
ssum 
*get,voktnwp,ssum,,item,volwp 
 
/com 
/com Plot commands for GWI 
/com 
cmsel,s,okngw 
nsle 
pfrac=0.1*nint(1000*(vokngw/vgw)) 
insul(1)=pfrac 
/title,wpbeam2%rn%,GW Insul., %pfrac% percent Pass Normal Stress Criteria 
plet,sx,avg 
cmsel,s,gw 
cmsel,u,okngw 
nsle 
ffrac=0.1*nint(1000*(1-vokngw/vgw)) 
/title,wpbeam2%rn%,GW Insul., %ffrac% percent Fail Normal Stress Criteria 
plet,sx,avg 
 
cmsel,s,oktgw 
nsle 
pfrac=0.1*nint(1000*(voktgw/vgw)) 
insul(2)=pfrac 
/title,wpbeam2%rn%,GW Insul., %pfrac% percent Pass Shear Stress Criteria 
plet,sxyxz,avg 
cmsel,s,gw 
cmsel,u,oktgw 
nsle 
ffrac=0.1*nint(1000*(1-voktgw/vgw)) 
/title,wpbeam2%rn%,GW Insul., %ffrac% percent Fail Shear Stress Criteria 
plet,sxyxz,avg 
 
cmsel,s,oktngw 
nsle 
pfrac=0.1*nint(1000*(voktngw/vgw)) 
insul(3)=pfrac 
/title,wpbeam2%rn%,GW Insul., %pfrac% percent Pass Shear & Normal Stress 
Criteria 
!/edge,1,1 
!eplo 
plet,sx,avg 
cmsel,s,gw 
cmsel,u,oktngw 
nsle 
ffrac=0.1*nint(1000*(1-voktngw/vgw)) 
/title,wpbeam2%rn%,GW Insul., %ffrac% percent Fail Shear & Normal Stress 
Criteria 
plet,sx,avg 
!eplo 
!/edge 
 
/com 
/com Plot commands for WPI 
/com 

cmsel,s,oknwp 
nsle 
pfrac=0.1*nint(1000*(voknwp/vwp)) 
insul(4)=pfrac 
/title,wpbeam2%rn%,WP Insul., %pfrac% percent Pass Normal Stress Criteria 
plet,sx,avg 
cmsel,s,wp 
cmsel,u,oknwp 
nsle 
ffrac=0.1*nint(1000*(1-voknwp/vwp)) 
/title,wpbeam2%rn%,WP Insul., %ffrac% percent Fail Normal Stress Criteria 
plet,sx,avg 
 
cmsel,s,oktwp 
nsle 
pfrac=0.1*nint(1000*(voktwp/vwp)) 
insul(5)=pfrac 
/title,wpbeam2%rn%,WP Insul., %pfrac% percent Pass Shear Stress Criteria 
plet,sxyxz,avg 
cmsel,s,wp 
cmsel,u,oktwp 
nsle 
ffrac=0.1*nint(1000*(1-voktwp/vwp)) 
/title,wpbeam2%rn%,WP Insul., %ffrac% percent Fail Shear Stress Criteria 
plet,sxyxz,avg 
 
cmsel,s,oktnwp 
nsle 
pfrac=0.1*nint(1000*(voktnwp/vwp)) 
insul(6)=pfrac 
/title,wpbeam2%rn%,WP Insul., %pfrac% percent Pass Shear & Normal Stress 
Criteria 
!/edge,1,1 
!eplo 
plet,sx,avg 
cmsel,s,wp 
cmsel,u,oktnwp 
nsle 
ffrac=0.1*nint(1000*(1-voktnwp/vwp)) 
/title,wpbeam2%rn%,WP Insul., %ffrac% percent Fail Shear & Normal Stress 
Criteria 
!eplo 
!/edge 
plet,sx,avg 
 
esel,s,mat,,3,5 
nsle 
smult,sy2,sy,sy 
smult,sz2,sz,sz 
sadd,sy2z2,sy2,sz2 
sexp,s_inpln,sy2z2,,0.5 
 
/title,wpbeam2%rn%,WP Insul., Normal Stresses 
plet,sx,avg 
/title,wpbeam2%rn%,WP Insul., In-Plane Stresses 
plet,s_inpln,avg 
allsel 
/title,wpbeam2%rn%, Deformation & Stiffness of TF WP 

/type 
 
/com 
/com Deformation 
/com 
allsel 
PATH,p1,2,30,nez+1, 
PPATH,1,0,0,0,0,0,   
PPATH,2,0,0,0,l_tot,0,   
PDEF,U,U,sum,AVG  
PLPATH,U    
PRPATH,U    
 
/com Write the Insulation Stress Criteria Review (% Passing Criteria) 
*vwrite, 
('GrWrp(Norm)  GrWrp(Tau)  GrWrp(N+T)   WP(Norm)    WP(Tau)     WP(N+T)') 
*vwrite,insul(1),insul(2),insul(3),insul(4),insul(5),insul(6) 
(1p6e12.4) 
 
fini 
/exit,all 
/eof 

6.5 ANSYS Input File, Smeared WP Beam 
 
/batch 
rn=0 
/filnam,beam1%rn% 
/show,beam1%rn%,grp 
!resume 
!*if,1,eq,1,:1000 
/prep7 
/com 
/com Structural model of quarter-Symmetry Fixed-Fixed Beam to benchmark smeared 
properties 
/com  of the NCSX TF WP with a slip-plane. 
/com 
/com Run History 
/com 
/com 10: Cross-Plane Shear adjusted to match detailed model deformations 
/com 11: All properties as calculated by smear1.dat (too stiff in bending) 
 
/com 
/com Misc Parameters 
/com 
l_tot=0.30               ! beam length 
em_pres=1e6              ! magnetic pressure loading 
k=0.0254                 ! english to si conversion factor 
 
/com 
/com graphics keys 
/com 
/pnum,mat,1 
/num,1 
/dist 
/focus 
/vup,1,x 
/view,1,,-1 
 
/com 
/com tf conductor, ignoring radii (in local CS vernacular) 
/com 
n_lay=3                   ! number of radial layers 
n_pan=4                   ! number of pancakes 
dr_con=0.966*k            ! Conductor build in thickness 
dz_con=0.709*k            ! Conductor build in height 
di_con=0.312*k            ! inside diameter of cooling channel 
 
/com 
/com Insulation 
/com 
t_tw=0.049*k              ! turn wrap insulation thk 
t_pan=0.0*k               ! pancake insulation thk (can be 0.0) 
t_lay=0.0*k               ! layer insulation thk (can be 0.0) 
t_gw=0.25*k               ! module over-wrap thickness 
t_gap=1e-5                ! Conductor/TW gap 
 
/com 
/com WP Build 
/com 
dr_wp=n_lay*(dr_con+2*t_tw)+(n_lay-1)*t_lay   ! radial build of WP 

dz_wp=n_pan*(dz_con+2*t_tw)+(n_pan-1)*t_pan   ! toroidal build of WP 
dr_iwp=dr_wp+2*t_gw                           ! radial build of ground-wrapped 
WP 
dz_iwp=dz_wp+2*t_gw                           ! toroidal build of ground-wrapped 
WP 
 
/com 
/com Smeared Material Properties 
/com 
 
mp,  ex,1,46.6e9 
mp,  ey,1,45.1e9 
mp,  ez,1,83.3e9 
mp,alpx,1,8.8e-6 
mp,alpy,1,9.1e-6 
mp,alpz,1,13.2e-6 
mp,nuxy,1,0.354 
mp,nuyz,1,0.294 
mp,nuxz,1,0.294 
mp,gxy,1,1.8e9 
mp,gyz,1,1.8e9   ! required to get beam bending right 
mp,gxz,1,1.8e9   ! required to get beam bending right 
 
/com 
/com element types 
/com 
et,1,45                  ! structure only 
 
/com 
/com Homogeneous WP 
/com 
cel=dr_iwp/2/5 
csys 
wpcsys 
block,-dr_iwp/2,dr_iwp/2,,dz_iwp/2,,l_tot 
lsel,s,length,,dz_iwp/2 
lesize,all,,,nint(dz_iwp/2/cel),,1 
lsel,s,length,,dr_iwp 
lesize,all,,,nint(dr_iwp/cel),,1 
lsel,s,length,,l_tot 
lesize,all,,,nint(l_tot/cel),,1 
allsel 
vmesh,all 
 
 
/com 
/com Symmetric BCs 
/com 
nsel,s,loc,y 
d,all,uy 
nsel,s,loc,z 
d,all,uz 
allsel 
nsel,s,loc,z,l_tot 
cp,next,uz,all 
nsel,s,loc,x,-dr_iwp/2 
nsel,r,loc,z,,!cel 
d,all,ux 

/com 
/com Apply a Surface Load 
csys 
esel,all 
nsel,s,loc,x,dr_iwp/2 
sf,all,pres,em_pres 
 
/title,beam1%rn%, Bending Def of Smeared TFWP 
allsel 
eplo 
allsel 
save 
fini 
:100 
/solu 
tref,0 
bfunif,temp,dtmp 
solve 
fini 
 
:1000 
/auto 
/post1 
 
/auto 
/view,1,1,1,1 
/title,beam1%rn%, Deformed Shape 
esel,s,mat,,2 
nsle 
pldi 
plns,s,int 
 
/com 
/com Deformation 
/com 
allsel 
PATH,p1,2,30,58, 
PPATH,1,0,0,0,0,0,   
PPATH,2,0,0,0,l_tot,0,   
PDEF,U,U,sum,AVG  
PLPATH,U    
PRPATH,U    
 
fini 
/exit,all 
/eof 

